# $Id: PKGBUILD 170191 2012-11-04 17:11:22Z giovanni $
# Maintainer:
# Contributor: damir <damir@archlinux.org>

pkgname=libnet
pkgver=1.1.6
pkgrel=1
pkgdesc="A library which provides API for commonly used low-level net functions"
arch=('i686' 'x86_64')
url="http://sourceforge.net/projects/libnet-dev/"
license=('BSD')
depends=('glibc' 'sh')
options=('!libtool')
source=("http://downloads.sourceforge.net/libnet-dev/${pkgname}-${pkgver}.tar.gz")
md5sums=('710296fe424a49344e5fcc0d09e53317')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr
  make CFLAGS="${CFLAGS} -fPIC"
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install

  install -Dm644 doc/COPYING \
    "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
